Exercises for a Friendly Guide to Mathematical Methods of Physics

```
In[ ]:= $PrePrint = TraditionalForm
```
Out[ ]= TraditionalForm

```
In[ ]:= Clear["Global`*"]
```

Complex Numbers

1. Compute the following powers of the imaginary unit:

i^{125}, i^{34}, $\frac{1}{i^5}$, i^{-15}

This can be easily done evaluating the following cell (the imaginary unit can be entered as ⋮ii⋮ *) wherein the elements of a list are inserted in curly braces:*

```
In[ ]:= {ⅈ^125, ⅈ^34, 1/ⅈ^5, ⅈ^-15}
```
Out[ ]= $\{i, -1, -i, i\}$

The powers of the imaginary unit cycle through the sequence of i, -1, $-i$, 1. Using FullSimplify (Mathematica 3.0), *with the assumption that n belongs to* $\mathbb{Z}$

```
In[ ]:= Table[ⅈ^(4 n + k), {k, 0, 3}] // FullSimplify[#, n ∈ Integers] &
```
Out[ ]= $\{1, i, -1, -i\}$

so, if n is any integer, we have

$$i^n = i^{n \bmod 4}$$

where mod *represents the modulo operation. Using the built-in commands* Mod (Mathematica 1.0) *and* FullSimplify, *we may verify the above equality*

```
In[ ]:= FullSimplify[ⅈ^n == ⅈ^Mod[n, 4], n ∈ Integers]
```
Out[ ]= True

the built-in command Equal (==) *obviously returns* True *if the lefthand side is equal to the righthand side.*

2. Simplify the following expressions writing the result as a complex number in the form *a* + ib:

a) $(2 + 3\,i)\left(\frac{2 - i}{1 + 2\,i}\right)^2$

In[∘]:= $(2+3\,i)\left(\frac{2-i}{1+2\,i}\right)^2$

Out[∘]= $-2-3\,i$

b) $\frac{4-5\,i+2\,i^3}{(2-i)^2}$

In[∘]:= $\frac{4-5\,i+2\,i^3}{(2-i)^2}$

Out[∘]= $\frac{8}{5}-\frac{i}{5}$

c) $\frac{1+i}{\sqrt{3}-i}$

If we simply input

In[∘]:= $\frac{1+i}{\sqrt{3}-i}$

Out[∘]= $\frac{1+i}{\sqrt{3}-i}$

Mathematica returns the cell unevaluated. If we want the result as a complex number in the algebraic form we may apply ComplexExpand (Mathematica 2.0)

In[∘]:= $\frac{1+i}{\sqrt{3}-i}$ // ComplexExpand

Out[∘]= $-\frac{1}{4}+\frac{\sqrt{3}}{4}+i\left(\frac{1}{4}+\frac{\sqrt{3}}{4}\right)$

3. Find the real and imaginary part of the following complex numbers:

a) $a+i\,b$

Let's apply the built-in function ReIm (Mathematica 11) *which returns a list whose elements are the real part and the imaginary part, respectively. In the first case, to assume that a and b are real numbers it is necessary to apply* ComplexExpand

In[∘]:= `ReIm[a + i b] // ComplexExpand`

Out[∘]= $\{a, b\}$

b) $\frac{i}{1-i}\,\frac{1}{2+3\,i}$

In[∘]:= ReIm$\left[\frac{i}{1-i}\,\frac{1}{2+3\,i}\right]$

Out[∘]= $\left\{\frac{1}{26}, \frac{5}{26}\right\}$

or in postfix notation

In[ ]:= $\frac{i}{1-i}\ \frac{1}{2+3i}$ // ReIm

Out[ ]= $\left\{\frac{1}{26}, \frac{5}{26}\right\}$

This is equivalent to evaluate the following cell:

In[ ]:= $\left\{\text{Re}\left[\frac{i}{1-i}\ \frac{1}{2+3i}\right], \text{Im}\left[\frac{i}{1-i}\ \frac{1}{2+3i}\right]\right\}$

Out[ ]= $\left\{\frac{1}{26}, \frac{5}{26}\right\}$

4. Let $z = x + i\,y$. Express the given quantity in terms of x and y.

a) Re $\left(\frac{1}{z}\right)$, Im $\left(\frac{1}{z}\right)$

to assume that x and y are real numbers it is necessary to apply ComplexExpand:

In[ ]:= ReIm$\left[\frac{1}{x+i\,y}\right]$ // ComplexExpand

Out[ ]= $\left\{\frac{x}{x^2+y^2}, -\frac{y}{x^2+y^2}\right\}$

b) Im $(2\,z + 4\,z^* - 4\,i)$

The complex conjugate is obtained with Conjugate *(or* $■^*$ *):*

In[ ]:= Im[2 z + 4 z* - 4 i] // ComplexExpand

Out[ ]= $2\,y - 4$

c) Im $\left(z^2 + z^{*2}\right)$

In[ ]:= Im[z² + Conjugate[z]²] // ComplexExpand

Out[ ]= 0

5. Find the modulus and argument of each of the following complex numbers:

a) $\frac{2\,i}{3-4\,i}$

Let's use the command AbsArg (Mathematica 10.1)

In[ ]:= AbsArg$\left[\frac{2\,i}{3-4\,i}\right]$

Out[ ]= $\left\{\frac{2}{5}, \pi - \tan^{-1}\left(\frac{3}{4}\right)\right\}$

b) $\frac{1-2\,i}{1+i} + \frac{2-i}{1-i}$

In[ ]:=
```
AbsArg[(1 - 2 ⅈ)/(1 + ⅈ) + (2 - ⅈ)/(1 - ⅈ)]
```
Out[ ]= $\left\{\sqrt{2}, -\frac{\pi}{4}\right\}$

c) $a + i\, b$

Assuming that a and b are real numbers

In[ ]:=
```
ComplexExpand[AbsArg[a + ⅈ b], TargetFunctions → {Re, Im}]
```
Out[ ]= $\left\{\sqrt{a^2 + b^2}, \tan^{-1}(a, b)\right\}$

or in postfix notation (note the use of a pure function)

In[ ]:=
```
AbsArg[a + ⅈ b] // ComplexExpand[#, TargetFunctions → {Re, Im}] &
```
Out[ ]= $\left\{\sqrt{a^2 + b^2}, \tan^{-1}(a, b)\right\}$

6. Obtain an Argand diagram for the following complex numbers:

$1 - \sqrt{3}\, i;\; 2\, i;\; \frac{\sqrt{2}}{2} + \frac{\sqrt{2}}{2}\, i;\; -\sqrt{3} - i$

There are various ways to plot complex numbers in the Wessel-Argand-Gauss plane. For example, one can use the command **`ComplexListPlot`** (Mathematica 12)

In[ ]:=
```
pts = {1 - √3 ⅈ, 2 ⅈ, √2/2 + √2/2 ⅈ, -√3 - ⅈ}
```
Out[ ]= $\left\{1 - i\sqrt{3}, 2\, i, \frac{1 + i}{\sqrt{2}}, -\sqrt{3} - i\right\}$

In[ ]:=
```
ComplexListPlot[pts, Axes → True, AxesLabel → {"Re(z)", "Im(z)"},
  PlotStyle → {Red, PointSize[Large]}, AspectRatio → 1, GridLines → Automatic];
```

Among the many options, one can label the points with **`Callout`** (Mathematica 11)

In[ ]:=
```
ComplexListPlot[Callout[#, #] & /@ pts, Axes → True, AxesLabel → {"Re(z)", "Im(z)"},
  PlotStyle → {Red, PointSize[Large]}, AspectRatio → 1, GridLines → Automatic,
  AxesStyle → Directive[Black, Arrowheads[{0.0, 0.03}]]]
```

Out[ ]=



alternatively with Directive

In[ ]:=
```
ComplexListPlot[Callout[#, #] & /@ pts, Axes → True, AxesLabel → {"Re(z)", "Im(z)"},
  PlotStyle → Directive[Orange, AbsolutePointSize[6]], AspectRatio → 1,
  GridLines → Automatic, AxesStyle → Directive[Black, Arrowheads[{0.0, 0.03}]]]
```

Out[ ]=



It is possible to visualize the points with a mesh in polar coordinates:

In[ ]:=
```
ComplexListPlot[Callout[#, #] & /@ pts,
 PolarGridLines → {Table[(π / 12) k, {k, 0, 24}], Table[k, {k, 1, 20}]},
 PolarAxes → True, AxesLabel → {"Re(z)", "Im(z)"},
 PlotStyle → {Red, PointSize[Large]}]
```

Out[ ]=


it is possible to show the Cartesian form of the points

In[ ]:=
```
newpoints = Table[ⅇ^(ⅈ 2 π k/8), {k, 0, 7}]
```

Out[ ]= $\left\{1, e^{\frac{i\pi}{4}}, i, e^{\frac{3i\pi}{4}}, -1, e^{-\frac{3i\pi}{4}}, -i, e^{-\frac{i\pi}{4}}\right\}$

In[ ]:=
```
ComplexListPlot[Callout[#, ComplexExpand[#]] &/@newpoints,
 PolarGridLines → {Table[(π / 12) k, {k, 0, 24}], Table[k, {k, 1, 20}]},
 PolarAxes → True,
 AxesLabel → {"Re(z)", "Im(z)"},
 PlotStyle → Directive[Orange, AbsolutePointSize[6]],
 AxesStyle → Directive[Black, Arrowheads[{0.0, 0.03}]]]
```

Out[ ]=


7. Given the complex numbers $z = a + i\,b$ and $w = c + i\,d$ calculate:

a) $z + w$, b) $z - w$

```
Clear[z, w]
```

In[ ]:=
```
z = a + ⅈ b;
w = c + ⅈ d;
```

In[ ]:=
```
{z + w, z - w} // ComplexExpand
```

Out[ ]= $\{a + i\,(b + d) + c,\; a + i\,(b - d) - c\}$

c) $z\,w$

In[ ]:=
```
z w // ComplexExpand
```

Out[ ]= $i\,(a\,d + b\,c) + a\,c - b\,d$

d) z/w

In[ ]:=
```
z/w // ComplexExpand
```

Out[ ]= $$i\left(\frac{b\,c}{c^2 + d^2} - \frac{a\,d}{c^2 + d^2}\right) + \frac{a\,c}{c^2 + d^2} + \frac{b\,d}{c^2 + d^2}$$

e) $1/z$

In[ ]:=
```
1/z // ComplexExpand
```
Out[ ]= $\frac{a}{a^2+b^2}-\frac{i\,b}{a^2+b^2}$

f) z^*

In[ ]:=
```
Conjugate[z] // ComplexExpand
```
Out[ ]= $a-i\,b$

g) $|z|$

In[ ]:=
```
Abs[z] // ComplexExpand
```
Out[ ]= $\sqrt{a^2+b^2}$

h) Arg(z)

In[ ]:=
```
Arg[z] // ComplexExpand[#, TargetFunctions → {Re, Im}] &
```
Out[ ]= $\tan^{-1}(a,b)$

8. Compute the indicated powers:

a) $(1-i)^{10}$

In[ ]:=
```
(1 - i)^10
```
Out[ ]= $-32\,i$

b) $\left(-1+i\sqrt{3}\right)^4$

writing simply

In[ ]:=
```
(-1 + i √3)^4
```
Out[ ]= $\left(-1+i\sqrt{3}\right)^4$

the power is left unevaluated. To obtain the value, one can try **Expand** (Mathematica 1.0), **RootReduce** (Mathematica 3.0) *or* **FullSimplify**

In[ ]:=
```
(-1 + i √3)^4 // Expand
```
Out[ ]= $-8+8\,i\sqrt{3}$

c) $\left[\cos\left(\frac{\pi}{8}\right)+i\sin\left(\frac{\pi}{8}\right)\right]^{12}$

this time **Expand** *doesn't help:*

In[ ]:=
```
(Cos[π/8] + i Sin[π/8])^12 // Expand
```
Out[ ]= $\sin^{12}\left(\frac{\pi}{8}\right)+\cos^{12}\left(\frac{\pi}{8}\right)+12\,i\sin\left(\frac{\pi}{8}\right)\cos^{11}\left(\frac{\pi}{8}\right)-66\sin^2\left(\frac{\pi}{8}\right)\cos^{10}\left(\frac{\pi}{8}\right)-220\,i\sin^3\left(\frac{\pi}{8}\right)\cos^9\left(\frac{\pi}{8}\right)+$
$495\sin^4\left(\frac{\pi}{8}\right)\cos^8\left(\frac{\pi}{8}\right)+792\,i\sin^5\left(\frac{\pi}{8}\right)\cos^7\left(\frac{\pi}{8}\right)-924\sin^6\left(\frac{\pi}{8}\right)\cos^6\left(\frac{\pi}{8}\right)-792\,i\sin^7\left(\frac{\pi}{8}\right)\cos^5\left(\frac{\pi}{8}\right)+$
$495\sin^8\left(\frac{\pi}{8}\right)\cos^4\left(\frac{\pi}{8}\right)+220\,i\sin^9\left(\frac{\pi}{8}\right)\cos^3\left(\frac{\pi}{8}\right)-66\sin^{10}\left(\frac{\pi}{8}\right)\cos^2\left(\frac{\pi}{8}\right)-12\,i\sin^{11}\left(\frac{\pi}{8}\right)\cos\left(\frac{\pi}{8}\right)$

so let's try **RootReduce**:

In[ ]:=
```
(Cos[π/8] + ⅈ Sin[π/8])^12 // RootReduce
```
Out[ ]= $-i$

d) i^{2i}

We have a complex power z^{α} *with base and exponent given by complex numbers. This function is in general multiple-valued*. Mathematica *computes the principal value of a complex power. In order to obtain said principal value we have to evaluate the following cell:*

In[ ]:=
```
ⅈ^(2 ⅈ) // ComplexExpand
```
Out[ ]= $e^{-\pi}$

e) $(4\,i)^{1+i}$

In[ ]:=
```
(4 ⅈ)^(1+ⅈ) // ComplexExpand
```
Out[ ]= $-4\,e^{-\pi/2}\sin(\log(4)) + 4\,i\,e^{-\pi/2}\cos(\log(4))$

9. Verify that $z = -1 \pm 2\,i$ satisfies the following equation: $z^3 + z^2 + 3\,z - 5 = 0$

Let's define a variable called **pol** *for the first hand of the equation using the command* **Set (=)** (Mathematica 1)

In[ ]:=
```
pol = -5 + 3 z + z^2 + z^3
```
Out[ ]= $z^3 + z^2 + 3\,z - 5$

then use **ReplaceAll (/.)** (Mathematica 1) *to evaluate the polynomial in the given points*

In[ ]:=
```
pol /. z → {-1 - 2 ⅈ, -1 + 2 ⅈ}
```
Out[ ]= $\{0, 0\}$

Alternatively one can solve the equation with **Solve** (Mathematica 1)

In[ ]:=
```
Solve[-5 + 3 z + z^2 + z^3 == 0, z]
```
Out[ ]= $\{\{z \to -1 - 2\,i\}, \{z \to -1 + 2\,i\}, \{z \to 1\}\}$

In[ ]:=
```
-5 + 3 z + z^2 + z^3 == 0 /. % // Simplify
```
Out[ ]= {True, True, True}

10. Solve simbolically the following equations:

a) $2\,z = i\,(2 + 9\,i)$

In[ ]:=
```
Clear[z]
```

In[ ]:=
```
Solve[2 z == ⅈ (2 + 9 ⅈ), z] // Flatten
```
Out[ ]= $\left\{z \to -\frac{9}{2} + i\right\}$

b) $z - 2\,z^{*} + 7 - 6\,i = 0$

In[ ]:=
```
Solve[z - 2 z* + 7 - 6 ⅈ == 0, z] // Flatten
```
Out[ ]= $\{z \to 7 + 2\,i\}$

c) $|z| - z = 2 + i$

In[ ]:= `Solve[Abs[z] - z == 2 + i, z, Complexes]`

Out[ ]= $\{\}$

in this case `Solve` *is unable to return a solution, so let's use* `Reduce` (Mathematica 1.0)

In[ ]:= `Reduce[Abs[z] - z == 2 + i, z]`

Out[ ]= $z = -\frac{3}{4} - i$

d) $z^6 - z^3 - 2 = 0$

In[ ]:= `Solve[z^6 - z^3 - 2 == 0, z]`

Out[ ]= $\{\{z \to -1\}, \{z \to -\sqrt[3]{-2}\}, \{z \to \sqrt[3]{-1}\}, \{z \to -(-1)^{2/3}\}, \{z \to \sqrt[3]{2}\}, \{z \to (-1)^{2/3}\sqrt[3]{2}\}\}$

if the solutions are to be expressed with complex numbers in the cartesian form one may try

In[ ]:= `Solve[z^6 - z^3 - 2 == 0, z] // ComplexExpand // Flatten`

Out[ ]= $\left\{z \to -1, z \to -\frac{1}{2^{2/3}} - \frac{i\sqrt{3}}{2^{2/3}}, z \to \frac{1}{2} + \frac{i\sqrt{3}}{2}, z \to \frac{1}{2} - \frac{i\sqrt{3}}{2}, z \to \sqrt[3]{2}, z \to -\frac{1}{2^{2/3}} + \frac{i\sqrt{3}}{2^{2/3}}\right\}$

it's possible to have a list of values of the solutions with `SolveValues` (Mathematica 12.3)

In[ ]:= `SolveValues[z^6 - z^3 - 2 == 0, z] // ComplexExpand`

Out[ ]= $\left\{-1, -\frac{1}{2^{2/3}} - \frac{i\sqrt{3}}{2^{2/3}}, \frac{1}{2} + \frac{i\sqrt{3}}{2}, \frac{1}{2} - \frac{i\sqrt{3}}{2}, \sqrt[3]{2}, -\frac{1}{2^{2/3}} + \frac{i\sqrt{3}}{2^{2/3}}\right\}$

to obtain the solutions in polar form one can use `PowerExpand`

In[ ]:= `Solve[z^6 - z^3 - 2 == 0, z] // Flatten // PowerExpand[#, Assumptions → True] &`

Out[ ]= $\left\{z \to -1, z \to -\sqrt[3]{-2}, z \to e^{\frac{i\pi}{3}}, z \to -e^{\frac{2i\pi}{3}}, z \to \sqrt[3]{2}, z \to \sqrt[3]{2}\, e^{\frac{2i\pi}{3}}\right\}$

in this case also it's possible to have only a list of the values

In[ ]:= `SolveValues[z^6 - z^3 - 2 == 0, z] // PowerExpand[#, Assumptions → True] &`

Out[ ]= $\left\{-1, -\sqrt[3]{-2}, e^{\frac{i\pi}{3}}, -e^{\frac{2i\pi}{3}}, \sqrt[3]{2}, \sqrt[3]{2}\, e^{\frac{2i\pi}{3}}\right\}$

e) $e^z = 2i$

This exponential equation has infinite solutions expressed as a conditional expression:

In[ ]:= `Solve[e^z == 2 i, z, Complexes] // Flatten`

Out[ ]= $\left\{z \to 2i\pi c_1 + \log(2) + \frac{i\pi}{2} \text{ if } c_1 \in \mathbb{Z}\right\}$

In order to select some solutions, e.g. for $c_1 = 0, 1, 2$, *one may try*

In[ ]:= `% /. C[1] → {0, 1, 2}`

Out[ ]= $\left\{z \to \left\{\log(2) + \frac{i\pi}{2}, \log(2) + \frac{5i\pi}{2}, \log(2) + \frac{9i\pi}{2}\right\}\right\}$

f) $\sin z = 2$

To solve this trigonometric equation let's solve the following equivalent equation:

In[∘]:=
```
SolveValues[(e^(i z) - e^(-i z))/(2 i) == 2, z, Complexes] // ComplexExpand
```

Out[∘]= $\left\{ -2\pi c_1 + i\log(2+\sqrt{3}) + \frac{\pi}{2} \text{ if } c_1 \in \mathbb{Z},\ -2\pi c_1 + i\log(2-\sqrt{3}) + \frac{\pi}{2} \text{ if } c_1 \in \mathbb{Z} \right\}$

let's find the two solutions for $c_1 = 0$

In[∘]:=
```
% /. C[1] → 0
```

Out[∘]= $\left\{\frac{\pi}{2} + i\log(2+\sqrt{3}), \frac{\pi}{2} + i\log(2-\sqrt{3})\right\}$

and a numerical approximation with `N` (Mathematica 1.0)

In[∘]:=
```
N[%]
```

Out[∘]= $\{1.5708 + 1.31696\, i, 1.5708 - 1.31696\, i\}$

11. Sketch the following sets in the complex plane:

a) $\{z \in \mathbb{C} : |z-(1+i)| = 2\}$; *b*) $\{z \in \mathbb{C} : |z-(1+i)| \le 2\}$

in the first case (circumference) the right command is `ComplexContourPlot` (Mathematica 12.1) while *in the second case (circle) the right command is* `ComplexRegionPlot` (Mathematica 12.1)

In[∘]:=
```
a = ComplexContourPlot[Abs[z - (1 + i)] == 2, {z, 3}, Axes → True,
    Frame → False, AxesLabel → {"Re(z)", "Im(z)"}, AspectRatio → Automatic];
```

In[∘]:=
```
b = ComplexRegionPlot[Abs[z - (1 + i)] ≤ 2, {z, 3}, Axes → True,
    Frame → False, AxesLabel → {"Re(z)", "Im(z)"}, AspectRatio → Automatic];
```

In order to show the two plots together one may use `GraphicsRow` (Mathematica 6.0)

In[∘]:=
```
GraphicsRow[{a, b}, ImageSize → Medium]
```

Out[∘]=


c) $|\text{Im}(z)| < |\text{Re}(z)|$

```
In[•]:= ComplexRegionPlot[Abs[Im[z]] < Abs[Re[z]], {z, 1}, Axes → True,
         Frame → False, AxesLabel → {"Re(z)", "Im(z)"}, PlotStyle → LightYellow,
         BoundaryStyle → Dashed, AxesStyle → Directive[Black, Arrowheads[{0.0, 0.03}]]]
```

Out[•]=


The style of the boundary is dashed because the boundary does not belong to the set (the set is open).

d) $\frac{\mathrm{Re}(z)}{|z-1|} > 1, \ \mathrm{Im}(z) < 2$

We have the intersection of two regions so we have to use the And *(&&) operator:*

In[◦]:=
```
ComplexRegionPlot[Re[z]/Abs[z - 1] > 1 && Im[z] < 2, {z, 1/4 - 5 i, 10 + 3 i}, Axes → True,
  Frame → False, AxesLabel → {"Re(z)", "Im(z)"}, PlotStyle → LightBlue,
  BoundaryStyle → Dashed, AxesStyle → Directive[Black, Arrowheads[{0.0, 0.03}]]]
```

Out[◦]=



we obtain the region interior to parabola $y^2 = 2(x - 1/2)$ *but below the line* $y = 2$.

12. Sketch the following sets in the complex plane. Moreover determine if the set is *open*, *closed*, a *domain*, *bounded* and *connected*.

a) $1 < |z - 1 - i| < 2$

The command is again `ComplexRegionPlot`

In[•]:=

```
ComplexRegionPlot[1 < Abs[z - 1 - ⅈ] < 2, {z, -2 - 2 ⅈ, 3 + 3 ⅈ}, Axes → True,
 Frame → False, AxesLabel → {"Re(z)", "Im(z)"}, PlotStyle → LightGreen,
 BoundaryStyle → Dashed, AxesStyle → Directive[Black, Arrowheads[{0.0, 0.03}]]]
```

Out[•]=



As one can easily see, the set is the interior of a circular ring and it’s open, connected (therefore a domain) and bounded.

b) $|\operatorname{Re}(z)| > 2$

In[•]:=

```
ComplexRegionPlot[Abs[Re[z]] > 2, {z, 4}, Axes → True,
 Frame → False, AxesLabel → {"Re(z)", "Im(z)"}, PlotStyle → LightRed,
 BoundaryStyle → Dashed, AxesStyle → Directive[Black, Arrowheads[{0.0, 0.03}]]]
```

Out[•]=



In this case the set is open, not connected, not a domain and unbounded.

13. Find all the values of the following roots and plot them in the Wessel-Argand plane. In case of roots of unity, verify that their sum is equal to 0:

a) $1^{\frac{1}{3}}$

the simplest way to compute the roots is to use the Inline Free-form Input and write: = all third roots of 1

In[∘]:=
```
ComplexRoots[1, 3]
```

Out[∘]= $\left\{e^{\frac{2 i \pi}{3}}, e^{-\frac{2 i \pi}{3}}, 1\right\}$

which is equivalent to call the **ComplexRoots** *function from the Wolfram Function Repository*

In[∘]:=
```
ResourceFunction["ComplexRoots"][1, 3]
```

Out[∘]= $\left\{e^{\frac{2 i \pi}{3}}, e^{-\frac{2 i \pi}{3}}, 1\right\}$

to convert the numbers into the algebraic form

In[∘]:=
```
% // ComplexExpand
```

Out[∘]= $\left\{-\frac{1}{2}+\frac{i\sqrt{3}}{2}, -\frac{1}{2}-\frac{i\sqrt{3}}{2}, 1\right\}$

In[∘]:=
```
% // RootOfUnityQ
```

Out[∘]= {True, True, True}

the above cell has been evaluated to confirm with `RootOfUnityQ` (Mathematica 6) *that the three complex numbers are roots of unity. In this case let's verify that their sum is zero using the function* `Total` (Mathematica 5.0):

In[∘]:=
```
ResourceFunction["ComplexRoots"][1, 3] // ComplexExpand // Total
```

Out[∘]= 0

If one is interested in the exponential and polar forms of the numbers, it is available in the Wolfram Function Repository the function **ComplexToPolar**

In[∘]:=
```
ResourceFunction["ComplexToPolar"][#, All] & /@
  ResourceFunction["ComplexRoots"][1, 3] // ComplexExpand
```

Out[∘]= $\left\{\left\langle\left|\text{Exponential} \to e^{\frac{2\pi}{3}\times i}, \text{Polar} \to 1\times\left(\cos\left(\frac{2\pi}{3}\right)+i\sin\left(\frac{2\pi}{3}\right)\right)\right|\right\rangle,\right.$
$\left\langle\left|\text{Exponential} \to e^{\left(-\frac{2\pi}{3}\right)\times i}, \text{Polar} \to 1\times\left(\cos\left(-\frac{2\pi}{3}\right)+i\sin\left(-\frac{2\pi}{3}\right)\right)\right|\right\rangle,$
$\left.\left\langle\left|\text{Exponential} \to e^{0\times i}, \text{Polar} \to 1\times(\cos(0)+i\sin(0))\right|\right\rangle\right\}$

to obtain the Argand diagram let's proceed as in Example 7. We obtain 3 points which are the vertexes of an equilateral triangle:

In[•]:=
```
Show[ComplexListPlot[ResourceFunction["ComplexRoots"][1, 3],
   PolarGridLines → {Table[(π / 12) k, {k, 0, 24}], Table[k, {k, 1, 20}]},
   PolarAxes → True, AxesLabel → {"Re(z)", "Im(z)"},
   PlotStyle → {Red, PointSize[Large]}],
  Table[Graphics[
    {Blue, Dashed, Line[{ReIm[Exp[ⅈ 2 π n / 3]], ReIm[Exp[ⅈ 2 π (n + 1) / 3]]}]}
   ], {n, 0, 2}]]
```

Out[•]=


b) $i^{\frac{1}{6}}$

In this case too, the roots are roots of unity

In[•]:=
```
radix = ResourceFunction["ComplexRoots"][ⅈ, 6]
```

Out[•]= $\left\{e^{\frac{5 i \pi}{12}}, e^{\frac{3 i \pi}{4}}, e^{-\frac{11 i \pi}{12}}, e^{-\frac{7 i \pi}{12}}, e^{-\frac{i \pi}{4}}, e^{\frac{i \pi}{12}}\right\}$

In[•]:=
```
radix // ComplexExpand // RootOfUnityQ
```

Out[•]= {True, True, True, True, True, True}

so let’s compute their sum as before

In[•]:=
```
radix // ComplexExpand // Total // Simplify
```

Out[•]= 0

the Argand diagram is made of 6 points which are the vertexes of a regular hexagon inscribed in the unit circle:

In[ ]:=
```
Show[
 ComplexListPlot[ResourceFunction["ComplexRoots"][ⅈ, 6],
  PolarGridLines → {Table[(π / 12) k, {k, 0, 24}], Table[k, {k, 1, 20}]},
  PolarAxes → True, AxesLabel → {"Re(z)", "Im(z)"},
  PlotStyle → {Blue, PointSize[Large]}],
 Table[Graphics[
   {Red, Dashed, Line[
     {ReIm[Exp[ⅈ π / 12] Exp[ⅈ π n / 3]], ReIm[Exp[ⅈ π / 12] Exp[ⅈ π (n + 1) / 3]]}]}
  ], {n, 0, 6}]]
```

Out[ ]=



Alternatively we can solve the corresponding equation:

In[ ]:=
```
sols = SolveValues[z^6 == ⅈ, z] // PowerExpand[#, Assumptions → True] &
```

Out[ ]= $\left\{-e^{\frac{i\pi}{12}}, e^{\frac{i\pi}{12}}, -e^{\frac{5i\pi}{12}}, e^{\frac{5i\pi}{12}}, -e^{\frac{3i\pi}{4}}, e^{\frac{3i\pi}{4}}\right\}$

obtaining the same result:

In[ ]:=
```
Sort @ ComplexExpand @ radix == Sort @ ComplexExpand @ sols // Simplify
```

Out[ ]= True

where `Sort` (Mathematica 1.0) *has been applied to compare the two lists.*

c) $(1+i)^{\frac{1}{3}}$

In this case the cubic roots

In[ ]:=
```
ResourceFunction["ComplexRoots"][1 + ⅈ, 3]
```

Out[ ]= $\left\{\sqrt[6]{2}\, e^{\frac{3i\pi}{4}}, \sqrt[6]{2}\, e^{-\frac{7i\pi}{12}}, \sqrt[6]{2}\, e^{\frac{i\pi}{12}}\right\}$

are not roots of unity

In[ ]:=
```
% // RootOfUnityQ
```
Out[ ]= {False, False, False}

In[ ]:=
```
ComplexListPlot[ResourceFunction["ComplexRoots"][1 + ⅈ, 3],
 PolarGridLines → {Table[(π / 12) k, {k, 0, 24}], Table[k, {k, 1, 20}]},
 PolarAxes → True, AxesLabel → {"Re(z)", "Im(z)"},
 PlotStyle → {Orange, PointSize[Large]}]
```

Out[ ]=


14. Find the first 15 terms of the following sequences of complex numbers

a) $\left\{\frac{i^{n-1}}{n}\right\}$

and check if the sequence is convergent or not.

It's easy to find a list of terms using `Table` (Mathematica 1.0)

In[ ]:=
```
Clear[succ]
```

In[ ]:=
```
succ[n_] := ⅈ^(n-1) / n
```

In[ ]:=
```
Table[succ[n], {n, 1, 15}]
```

Out[ ]= $\left\{1, \frac{i}{2}, -\frac{1}{3}, -\frac{i}{4}, \frac{1}{5}, \frac{i}{6}, -\frac{1}{7}, -\frac{i}{8}, \frac{1}{9}, \frac{i}{10}, -\frac{1}{11}, -\frac{i}{12}, \frac{1}{13}, \frac{i}{14}, -\frac{1}{15}\right\}$

In[ ]:=
```
ComplexListPlot[Table[succ[n], {n, 1, 15}],
 PolarGridLines → {Table[(π / 12) k, {k, 0, 24}], Table[k, {k, 1, 20}]},
 PolarAxes → True, AxesLabel → {"Re(z)", "Im(z)"},
 PlotStyle → {Red, PointSize[Medium]}, Joined → True]
```

Out[ ]=



let'use `DiscretePlot` (Mathematica 7.0)

```
DiscretePlot[{Re[ⅈ^(n+1)/n], Im[ⅈ^(n+1)/n]}, {n, 50},
 PlotLegends → {"Re(zₙ)", "Im(zₙ)"}, PlotRange → {-1.2, 0.5}]
```

Out[ ]=



graphically we see that both the real and imaginary parts approach 0. In fact

In[ ]:=
```
Limit[ⅈ^(n+1)/n, n → ∞]
```

Out[ ]= 0

In[∘]:= `Limit[ReIm[`$\frac{i^{n+1}}{n}$`], n → ∞]`

Out[∘]= $\{0, 0\}$

b) $\left\{\frac{n - i^n}{\sqrt{n}}\right\}$

In[∘]:= `Clear[succ]`

In[∘]:= `succ[n_] :=` $\frac{n - i^n}{\sqrt{n}}$

In[∘]:= `Table[succ[n], {n, 1, 15}]`

Out[∘]= $\left\{1 - i, \frac{3}{\sqrt{2}}, \frac{3 + i}{\sqrt{3}}, \frac{3}{2}, \frac{5 - i}{\sqrt{5}}, \frac{7}{\sqrt{6}}, \frac{7 + i}{\sqrt{7}}, \frac{7}{2\sqrt{2}}, 3 - \frac{i}{3}, \frac{11}{\sqrt{10}}, \frac{11 + i}{\sqrt{11}}, \frac{11}{2\sqrt{3}}, \frac{13 - i}{\sqrt{13}}, \frac{15}{\sqrt{14}}, \frac{15 + i}{\sqrt{15}}\right\}$

In[∘]:= `DiscretePlot[{Re[succ[n]], Im[succ[n]]},`
`{n, 50}, PlotLegends → {"Re(z)", "Im(z)"}]`

Out[∘]=


the sequence is divergent

In[∘]:= `Limit[`$\frac{n + i^n}{\sqrt{n}}$`, n → ∞]`

Out[∘]= ∞

In[∘]:= `Limit[ReIm[`$\frac{n + i^n}{\sqrt{n}}$`], n → ∞]`

Out[∘]= $\{\infty, 0\}$

b) $\left\{1 + \frac{1}{(1 + i)^n}\right\}$

In[∘]:= `Clear[succ]`

In[∘]:= `succ[n_] := 1 -` $\frac{1}{(1 + i)^n}$

In[•]:=
```
Table[succ[n], {n, 1, 15}]
```

Out[•]=
$$\left\{\frac{1}{2}+\frac{i}{2},\ 1+\frac{i}{2},\ \frac{5}{4}+\frac{i}{4},\ \frac{5}{4},\ \frac{9}{8}-\frac{i}{8},\ 1-\frac{i}{8},\ \frac{15}{16}-\frac{i}{16},\ \frac{15}{16},\ \frac{31}{32}+\frac{i}{32},\ 1+\frac{i}{32},\ \frac{65}{64}+\frac{i}{64},\ \frac{65}{64},\ \frac{129}{128}-\frac{i}{128},\ 1-\frac{i}{128},\ \frac{255}{256}-\frac{i}{256}\right\}$$

In[•]:=
```
ComplexListPlot[Table[{Labeled[succ[n], Subscript[z, n]]}, {n, 1, 15}],
 AxesLabel → {"Re(z)", "Im(z)"}, PlotRange → All, GridLines → Automatic,
 AxesStyle → Directive[Black, Arrowheads[{0.0, 0.03}]]]
```

Out[•]=


In[•]:=
```
DiscretePlot[{Re[succ[n]], Im[succ[n]]},
 {n, 40}, PlotLegends → {"Re(zn)", "Im(zn)"}]
```

Out[•]=


In[•]:=
```
Limit[succ[n], n → ∞]
```

Out[•]= 1

15. Show that the given sequence $\{z_n\}$ converges to a complex number L by computing the limits of the real and imaginary parts.

a) $\left\{\left(\frac{1+i}{2}\right)^n\right\}$

it's easy to compute $\lim_{n\to\infty} \operatorname{Re}(z_n)$

In[ ]:= $\texttt{Limit}\left[\texttt{Re}\left[\left(\frac{1+i}{2}\right)^n\right], n \to \infty\right]$

Out[ ]= 0

and $\lim_{n\to\infty} \operatorname{Im}(z_n)$

In[ ]:= $\texttt{Limit}\left[\texttt{Im}\left[\left(\frac{1+i}{2}\right)^n\right], n \to \infty\right]$

Out[ ]= 0

b) $\left\{\frac{2+n\,i}{n+2\,n\,i}\right\}$

In[ ]:=
```
Clear[succ]
```

In[ ]:= $\texttt{succ[n\_] :=} \frac{2+n\,i}{n+2\,n\,i}$

In[ ]:=
```
DiscretePlot[{Re[succ[n]], Im[succ[n]]}, {n, 50},
 PlotLegends → {"Re(zn)", "Im(zn)"}, PlotRange → {-0.8, 1.2}]
```

Out[ ]=


From the plot we see that the real and imaginary parts approach two different complex numbers. Both limits may be evaluated simultaneously with

In[ ]:= $\texttt{Limit}\left[\texttt{ReIm}\left[\frac{2+n\,i}{n+2\,n\,i}\right], n \to \infty\right]$

Out[ ]= $\left\{\frac{2}{5}, \frac{1}{5}\right\}$

In[ ]:= $\texttt{Limit}\left[\frac{2+n\,i}{n+2\,n\,i}, n \to \infty\right]$

Out[ ]= $\frac{2}{5} + \frac{i}{5}$

16. Determine whether the given geometric series is convergent or divergent. If convergent, find its sum

a) $\sum_{k=1}^{\infty}\left(\frac{i}{2}\right)^k$

A geometric series is any series of the form

$$\sum_{k=0}^{\infty} a\, z^k$$

convergent for $|z| < 1$ *and having for sum* $a/(1-z)$*. To compute directly its sum the command is* `Sum` (Mathematica 1.0) *used in this case with the option* `GenerateConditions→True`

In[◦]:= `Sum[a z^k, {k, 0, ∞}, GenerateConditions → True]`

Out[◦]= $\frac{a}{1-z}$ if $|z| < 1$

Alternatively to check the convergence it's useful `SumConvergence` (Mathematica 7.0)

In[◦]:= `SumConvergence[a z^k, k]`

Out[◦]= $a == 0 \lor |z| < 1$

the first n terms of the geometric series may be computed as

In[◦]:= `Sum[a z^k, {k, 0, n - 1}]`

Out[◦]= $\frac{a\,(z^n - 1)}{z - 1}$

In[◦]:= `Limit[%, n → ∞]`

Out[◦]= $-\frac{a}{z-1}$ if $\left(a \,\middle|\, \frac{1}{z-1}\right) \in \mathbb{R} \land \log(z) < 0$

The above series is a particular geometric series (a = 1 and z = i/2)

In[◦]:= `SumConvergence[(i/2)^k, k]`

Out[◦]= True

in fact $|i/2| = 1/2 < 1$*. Moreover the generic term tends to zero*

In[◦]:= `Limit[(i/2)^k, k → ∞]`

Out[◦]= 0

the partial sum is

In[◦]:= `Sum[(i/2)^k, {k, 1, n}]`

Out[◦]= $\left(\frac{1}{5} - \frac{2\,i}{5}\right)\left(-1 + \left(\frac{i}{2}\right)^n\right)$

In[◦]:= `Limit[%, n → ∞]`

Out[◦]= $-\frac{1}{5} + \frac{2\,i}{5}$

to compute directly the sum

In[•]:=
```
Sum[(ⅈ/2)^k, {k, 1, ∞}]
```

Out[•]= $-\frac{1}{5}+\frac{2\,i}{5}$

or, using ESC sum ESC *to enter* ∑ *and* CTRL _ *for the lower limit and then* CTRL % *for the upper limit:*

In[•]:= $\sum_{k=1}^{\infty}\left(\frac{i}{2}\right)^k$

Out[•]= $-\frac{1}{5}+\frac{2\,i}{5}$

ratio test

In[•]:=
```
Limit[Abs[(ⅈ/2)^(n+1) / (ⅈ/2)^n], n → ∞]
```

Out[•]= $\frac{1}{2}$

as the limit is less than 1 the series is absolutely convergent.

b) $\sum_{k=0}^{\infty}(1-2\,i)^k$

prove immediately that it's divergent

In[•]:=
```
Limit[(1 - 2 ⅈ)^k, k → ∞]
```

Out[•]= ComplexInfinity

In[•]:=
```
SumConvergence[(1 - 2 ⅈ)^k, k]
```

Out[•]= False

In[•]:=
```
Sum[(1 - 2 ⅈ)^k, {k, 0, ∞}]
```

Sum: Sum does not converge.

Out[•]= $\sum_{k=0}^{\infty}(1-2\,i)^k$

17. Given a real symbolic quaternion q find: the absolute value, the vector part, the conjugate quaternion, the absolute value and the norm

Let's first load the Quaternion Analysis Package

In[•]:=
```
Clear["Global`*"]
```

In[•]:=
```
<< QuaternionAnalysis`
```

SetCoordinates: -- Message text not found -- ({X0, X1, X2, X3})

In[•]:=
```
q = Quaternion[x0, x1, x2, x3]
```

Out[•]= $x_0+\mathbb{i}\,x_1+\mathbb{j}\,x_2+\mathbb{k}\,x_3$

In[∘]:= `{Re[q], Vec[q]}`

Out[∘]= $\{x_0, \text{Quaternion}(0, x_1, x_2, x_3)\}$

In[∘]:= `Vec[q]`

Out[∘]= $\mathbb{i}\, x_1 + \mathbb{j}\, x_2 + \mathbb{k}\, x_3$

In[∘]:= `Conjugate[q]`

Out[∘]= $x_0 - \mathbb{i}\, x_1 - \mathbb{j}\, x_2 - \mathbb{k}\, x_3$

In[∘]:= `Abs[q]`

Out[∘]= $\sqrt{x_0^2 + x_1^2 + x_2^2 + x_3^2}$

In[∘]:= `AbsVec[q]`

Out[∘]= $\sqrt{x_1^2 + x_2^2 + x_3^2}$

In[∘]:= `Norm[q]`

Out[∘]= $\sqrt{x_0^2 + x_1^2 + x_2^2 + x_3^2}$

In[∘]:= `q ** Conjugate[q]`

Out[∘]= $x_0^2 + x_1^2 + x_2^2 + x_3^2$

18.

19.

20.

21.

22.

23.

24.

25.

26.

27.

28.